unittest
{
	assert( [ "3", "2 0", "3 1", "1 3", "4 2", "0 4", "5 5" ].parse.expand.solve == 2 );
	assert( [ "3", "0 0", "1 1", "5 2", "2 3", "3 4", "4 5" ].parse.expand.solve == 2 );
	assert( [ "2", "2 2", "3 3", "0 0", "1 1" ].parse.expand.solve == 0 );
	assert( [ "5", "0 0", "7 3", "2 2", "4 8", "1 6", "8 5", "6 9", "5 4", "9 1", "3 7" ].parse.expand.solve == 5 );
	assert( [ "5", "0 0", "1 1", "5 5", "6 6", "7 7", "2 2", "3 3", "4 4", "8 8", "9 9" ].parse.expand.solve == 4 );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.string;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = input.front.strip.to!( long );
	
	auto rs = new long[ 2 ][ n ];
	foreach( ref r; rs )
	{
		input.popFront;
		r = input.front.split.to!( long[] );
	}
	
	auto bs = new long[ 2 ][ n ];
	foreach( ref b; bs )
	{
		input.popFront;
		b = input.front.split.to!( long[] );
	}
	
	return tuple( n, rs, bs );
}

auto solve( long n, long[ 2 ][] rs, long[ 2 ][] bs )
{
	// x座標順にソート
	rs.sort();
	bs.sort();
	
	// x座標が小さい青点から順に判定
	auto c = 0L;
	foreach( b; bs )
	{
		// 範囲内の赤点のうち、一番y座標が大きいものを検索
		auto rmi = -1L;
		foreach( i, r; rs )
		{
			if( r[ 0 ] < b[ 0 ] && r[ 1 ] < b[ 1 ] )
			{
				if( rmi < 0 || rs[ rmi ][ 1 ] < r[ 1 ] )
				{
					rmi = i;
				}
			}
		}
		
		// 範囲内に赤点があったらペア数+1して使用済赤点を削除
		if( 0 <= rmi )
		{
			c++;
			rs = rs[ 0 .. rmi ] ~ rs[ rmi + 1 .. $ ];
		}
	}
	
	return c;
}
